
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc153775863]Corrections required
1.1 [bookmark: _Toc153775864]Change to the Function f_NRDC_RRC_PSCellChange_InterCell_SplitMCGReestablish
	Function name
	f_NRDC_RRC_PSCellChange_InterCell_SplitMCGReestablish

	Reason for change
	In the current TTCN Implementation of the function f_NRDC_RRC_PSCellChange_InterCell_SplitMCGReestablish for the test case 8.2.3.14.2

(1) As per 38.523-1 it is expected for UE to receive an RRCReconfiguration message to handover from NR PSCell involving SCG change with reconfigurationWithSync to the target NR cell with Split DRB .but in the current TTCN implementation , on the Target Pscell Id , the split bearer configuration is not present when configuring v_SS_RadioBearerDRB_SCG. Instead it is configured as SCG DRB. this needs to be addressed and It is proposed to configure Split DRB .

(2) When configuring v_SS_MCG_RadioBearerList with Split SCG DRB, SDAP configuration is set to omit , but for SA this has to be set as SDAP_Configuration_None.

(3) v_CellGroupConfigMCG is modified to have reconfigurationwithSync  on MCG and random access configurations are provided accordingly in the MCG Pcell. v_CellGroupConfigMCG is assigned to have f_NR_GetCellGroupConfigHO instead of cs_38508_CellGroupConfig. This change is to be in line with other cases where TTCN is doing f_NR5GC_GetExisting_BearerToAddModList(true_); //rlc reestablish

(4) due to change(3), it is proposed to add f_NR_SS_C_RNTI_Config() and f_NR_SS_RachProcedureConfigHO() HO on MCG_PcellId along with f_NR_SS_SRBs_DRBs_Release by using tsc_FollowOnFlag
 
(5) In step 1 , target Pscell Id is configured with radio bearers ,especially as per the test purpose , we need to have Split DRB. but this configuration can lead to both source Spcell and Target Spcell containing 
   	   LinkToOtherCellGroup
			Value = NR (1)
				Value = nr_Cell1 (1) at the same time
              For SS , this can lead to two terminating PDCP entitieis referring to   the same LinkToOtherCellGroup.
To address this issue it is proposed to do the step “1. NR Target PSCell: Configuration of SRB3 (if necessary) and DRBs” once source Pscell radio bearers are released. This also means we need to PdcpCountHandOver from source to target Pscell & security key update on the Target Pscell once radio bearers are established.
Also it is proposed to configure bearers on p_Target_PSCellId, after the reception of reconfiguration complete , so that we don’t have too many time based actions during the reconfiguration procedure on MCG Cell.

In addition , To do the reconfigurationWithSync rach configuration on the target PScell with newUE_Identity = (16712) , tsc_C_RNTI_Value3 is to be configured . so that it matches with peer reconfiguration configuration.
This is done by adding an extra parameter to the function f_NR_GetRACHProcedureConfigPSCell

	Summary of change
	(1) v_SS_RadioBearerDRB_SCG is configured as Split DRB to match the test purpose requirement

(2) v_SS_MCG_RadioBearerList configuration to have Sdap set to None instead of Omit

(3) v_CellGroupConfigMCG:= f_NR_GetCellGroupConfigHO change is made so that TTCN is in line wherever TTCN is doing f_NR5GC_GetExisting_BearerToAddModList(true_); //rlc reestablish

(4) f_NR_SS_C_RNTI_Config() and f_NR_SS_RachProcedureConfigHO() are added on MCG_PcellId along with f_NR_SS_SRBs_DRBs_Release by using tsc_FollowOnFlag, for achieving reconfigurationWithSync procedure

(5) Changes are done to avoid two SCG Entities having the same LinkToOtherCellGroup configured at the same time

	TTCN module
	develop\Common\NR5GC\NR5GC_RRCSteps_NRDC.ttcn

	MCC160 Comment
	Accepted in principle, implemented as per R5-240547



Before Change:
	    function f_NRDC_RRC_PSCellChange_InterCell_SplitMCGReestablish( NR_CellId_Type    p_MCG_PCellId,
                                                                   NR_CellId_Type      p_Source_PSCellId,
                                                                   NR_CellId_Type      p_Target_PSCellId,
                                                                   boolean             p_RBConfig_KeyChange := true,
                                                                   RNTI_Value          p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                                   template (omit) SK_Counter  p_SK_Counter := omit
                                                                  ) runs on NR5GC_PTC
  {
    var integer v_SCG_DRB_Id := f_NR_GetDefaultDRB_SCG();
    var template (value) SDAP_Config v_SDAP_ConfigSCG;
    var template (value) SDAP_Configuration_Type v_SDAP_ConfigurationSCG;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN);
    var template (value) NR_RadioBearer_Type v_SS_RadioBearerDRB_SCG;
    var template (value) NR_RadioBearerList_Type v_SS_TargetSCG_RadioBearerList;
    var template (value) NR_RadioBearerList_Type v_SourceSCG_RadioBearerToReleaseList := {cs_NR_SS_DRB_Release(v_SCG_DRB_Id)};
    var template (value) CellGroupConfig v_CellGroupConfigSCG;
    var template (value) RadioBearerConfig v_RadioBearerConfig2;
    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerToAddModListMCG:= f_NR5GC_GetExisting_BearerToAddModList(true_);
    var template (value) CellGroupConfig v_CellGroupConfigMCG:= cs_38508_CellGroupConfig(tsc_NR_CellGroupId_MCG, v_RLC_BearerToAddModListMCG );
    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig := f_NR_GetRACHProcedureConfigPSCell( p_Target_PSCellId, -);
    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get();
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var IntegerList_Type v_SS_DrbToReleaseMCG := f_NR_GetActiveDRBs(p_MCG_PCellId);
    var template (value) NR_DRBInfoList_Type v_MCG_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();
    var template (omit) NR_RadioBearerList_Type v_SS_MCG_RadioBearerList := f_NR5GC_GetRadioBearerList(v_MCG_DRBInfoList, true);
    var boolean v_IsFR1_SCG := f_NR_CellInfo_GetIsFR1(p_Target_PSCellId);
    var SubFrameTiming_Type v_Timing;
        
    //To include in RRCReconfiguration
    //Get SDAP - @sic R5s211604 sic@
    if (isvalue(v_GSM_MobilityInfo)) {
      v_SDAP_ConfigSCG := cs_SDAP_Config(oct2int(valueof(v_GSM_MobilityInfo.SessionId)), false, {5});
      v_SDAP_ConfigurationSCG:= f_ReturnNR_SDAP_Configuration_FromSDAP_Config(valueof(v_SDAP_ConfigSCG));
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No internet PDU session found to modify");
    }
    v_SS_RadioBearerDRB_SCG := f_NR_GetRadioBearerConfig_FullAM(p_Target_PSCellId, v_SCG_DRB_Id,-,-,-,v_SDAP_ConfigurationSCG);
    v_SS_TargetSCG_RadioBearerList := {v_SS_RadioBearerDRB_SCG};
    v_CellGroupConfigSCG := f_NR_GetSCGConfigNRDC_SCG(p_Target_PSCellId, v_SCG_DRB_Id, p_RNTI_Value);
    v_CellGroupConfigSCG.rlc_BearerToAddModList[0].reestablishRLC := true_;
    //@sic R5s221057 sic@
    v_RadioBearerConfig2 := cs_NRDRB_SplitBearer(omit, v_SCG_DRB_Id, cs_NR_PDCP_Config_Split(cs_38508_PDCP_Config_DRB, omit, tsc_NR_CellGroupId_SCG), secondary, f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get(), true_);
    v_V1530Ext := f_NRDC_BuildRRCReconfiguration_v1530( v_CellGroupConfigSCG, v_CellGroupConfigMCG, v_RadioBearerConfig2, -, p_SK_Counter );
    
    // Add SCG DRB configured in proxy (split SCG DRB)
    v_SS_MCG_RadioBearerList[lengthof(v_SS_MCG_RadioBearerList)] := cs_NR_SS_DRB_Config(v_SCG_DRB_Id, cs_NR_PDCP_Configuration_Proxy(cs_RlcBearerRouting_NR(p_Target_PSCellId)), cs_NR_RlcBearerConfig_AM(v_IsFR1_SCG, -, f_NR_LogicalChannelId_DRB(v_SCG_DRB_Id)));


    //1. NR Target PSCell: Configuration of SRB3 (if necessary) and DRBs
    f_NR_SS_CommonRadioBearerConfig(p_Target_PSCellId, v_SS_TargetSCG_RadioBearerList);
    
    // 2. Transfer of the PDCP Count for DRBs and SRB3 (if necessary) from NR source to NR target PSCell:
    //a) NR Source PSCell: Get PDCP COUNT.
    //b) NR Target PSCell: Set PDCP COUNT.
    //NOTE 1: No further sending/receiving of DRB data before the PSCell change is done.
    //NOTE 2: For AM DRBs the PDCP count is maintained. For SRB3 (if applied) and UM DRBs, the PDCP count is maintained or reset depending on the RRCReconfiguration message content.
    f_NR_SS_PdcpCount_Handover(p_Source_PSCellId, p_Target_PSCellId, v_SS_TargetSCG_RadioBearerList, p_RBConfig_KeyChange);
    
    //3. NR Target PSCell: Inform the SS about the PSCell change and about the source cell id.
    f_NR_SS_PdcpHandoverCtrl(p_Target_PSCellId, cas_NR_PdcpHandoverInit_REQ(p_Source_PSCellId, p_Target_PSCellId));
        
    //4. NR Target PSCell: Configure RACH procedure either dedicated or C-RNTI based.
    //NOTE 3: The FollowOnFlag is set to true in the ASP reconfiguring C-RNTI.
    f_NR_CellInfo_SetRNTI(p_Target_PSCellId, p_RNTI_Value);
    f_NR_SS_C_RNTI_Config(p_Target_PSCellId, p_RNTI_Value, -, tsc_NoCnfReq, tsc_FollowOnFlag);
    f_NR_SS_RachProcedureConfigHO(p_Target_PSCellId, v_RachProcedureConfig);
    
    //5. NR Target PSCell: Activate security.
    if (isvalue(p_SK_Counter)) {
        v_SecurityParams.SK_Counter := valueof(p_SK_Counter);
        v_SecurityParams := f_NR_Security_InitSCG(v_SecurityParams);
        f_NR_Security_Set(v_SecurityParams);
    }
    v_SecurityParams.AS_Ciphering_S.ActTimeList := f_NR_RRC_CipherActTime_SCG(p_Target_PSCellId, v_SS_TargetSCG_RadioBearerList);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_Target_PSCellId,
                                          v_SecurityParams.AS_Integrity_S,
                                          v_SecurityParams.AS_Ciphering_S);
                                          
    //6. NR Target PSCell: Configure UL grant configuration ("OnSR", default grant).
    //NOTE 3: Unless explicitly specified UL grant configuration keeps configured as per default at the NR source cell.
    f_NR_ULGrantConfiguration_Start(p_Target_PSCellId, -, -, -, -, cs_UL_GrantConfig_OnSR);
 
    //7. NR PCell: Send RRCReconfiguration.
    //8. NR PCell: Receive RRCReconfigurationComplete.
    v_Timing := f_NR_GetNextSendOccasion_DL(p_MCG_PCellId);
    f_NR_SendRRCReconfiguration (p_MCG_PCellId, tsc_NR_RbId_SRB1, -, v_V1530Ext, -, -, cs_TimingInfo_NR(v_Timing), false);
    
    //Reset RLC in MCG
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);
    f_NR_SS_SRBs_DRBs_Release(p_MCG_PCellId, cs_TimingInfo_NR(v_Timing), v_SS_DrbToReleaseMCG);
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms
    f_NR_SS_SRBs_DRBs_Config(p_MCG_PCellId, cs_TimingInfo_NR(v_Timing), v_SS_MCG_RadioBearerList);

    SRB.receive(car_NR_SRB_RrcPdu_IND(p_MCG_PCellId, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete(cr_RRCReconfigurationComplete_v1530(-, cr_RRCReconfigurationComplete_v1560(?)))));

    //9. NR Target PSCell: Inform the SS about completion of the PSCell change (e.g. to trigger PDCP STATUS REPORT PDU).
    f_NR_SS_PdcpHandoverCtrl(p_Target_PSCellId, cas_NR_PdcpHandoverComplete_REQ(p_Target_PSCellId));

    //10. NR Source PSCell: Release SRB3 (if necessary) and DRBs.
    f_NR_SS_CommonRadioBearerConfig(p_Source_PSCellId, v_SourceSCG_RadioBearerToReleaseList);
   }



After Change:
	  function f_NRDC_RRC_PSCellChange_InterCell_SplitMCGReestablish( NR_CellId_Type    p_MCG_PCellId,
                                                                   NR_CellId_Type      p_Source_PSCellId,
                                                                   NR_CellId_Type      p_Target_PSCellId,
                                                                   boolean             p_RBConfig_KeyChange := true,
                                                                   RNTI_Value          p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                                   template (omit) SK_Counter  p_SK_Counter := omit
                                                                  ) runs on NR5GC_PTC
  {
    var integer v_SCG_DRB_Id := f_NR_GetDefaultDRB_SCG();
    var template (value) SDAP_Config v_SDAP_ConfigSCG;
    var template (value) SDAP_Configuration_Type v_SDAP_ConfigurationSCG;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN);
    var template (value) NR_RadioBearer_Type v_SS_RadioBearerDRB_SCG;
    var template (value) NR_RadioBearerList_Type v_SS_TargetSCG_RadioBearerList;
    var template (value) NR_RadioBearerList_Type v_SourceSCG_RadioBearerToReleaseList := {cs_NR_SS_DRB_Release(v_SCG_DRB_Id)};
    var template (value) CellGroupConfig v_CellGroupConfigSCG;
    var template (value) RadioBearerConfig v_RadioBearerConfig2;
    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerToAddModListMCG:= f_NR5GC_GetExisting_BearerToAddModList(true_);
    //var template (value) CellGroupConfig v_CellGroupConfigMCG:= cs_38508_CellGroupConfig(tsc_NR_CellGroupId_MCG, v_RLC_BearerToAddModListMCG ); //WA#WI=970303
    var template (value) CellGroupConfig v_CellGroupConfigMCG:= f_NR_GetCellGroupConfigHO (nr_Cell1,v_RLC_BearerToAddModListMCG );  //WA#WI=970303
    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig := f_NR_GetRACHProcedureConfigPSCell( p_Target_PSCellId, -,p_RNTI_Value);//WA#WI=970303);
    var NR_SecurityParams_Type v_SecurityParams := f_NR_Security_Get();
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var IntegerList_Type v_SS_DrbToReleaseMCG := f_NR_GetActiveDRBs(p_MCG_PCellId);
    var template (value) NR_DRBInfoList_Type v_MCG_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();
    var template (omit) NR_RadioBearerList_Type v_SS_MCG_RadioBearerList := f_NR5GC_GetRadioBearerList(v_MCG_DRBInfoList, true);
    var boolean v_IsFR1_SCG := f_NR_CellInfo_GetIsFR1(p_Target_PSCellId);
    var SubFrameTiming_Type v_Timing;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_MCG_PCellId); //WA#WI=970303
    var template (value) NR_RachProcedureConfig_Type  v_RachProcedureConfig_MCG := cs_NR_RachProcedureConfig_CRNTI_HO(v_NR_PhysicalParameters, tsc_C_RNTI_Value3);//WA#WI=970303
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;   //WA#WI=970303
        
    //To include in RRCReconfiguration
    //Get SDAP - @sic R5s211604 sic@
    if (isvalue(v_GSM_MobilityInfo)) {
      v_SDAP_ConfigSCG := cs_SDAP_Config(oct2int(valueof(v_GSM_MobilityInfo.SessionId)), false, {5});
      v_SDAP_ConfigurationSCG:= f_ReturnNR_SDAP_Configuration_FromSDAP_Config(valueof(v_SDAP_ConfigSCG));
    } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No internet PDU session found to modify");
    }
    //v_SS_RadioBearerDRB_SCG := f_NR_GetRadioBearerConfig_FullAM(p_Target_PSCellId, v_SCG_DRB_Id,-,-,-,v_SDAP_ConfigurationSCG); //WA#WI=970303
   
    v_SS_RadioBearerDRB_SCG := cs_NR_SS_DRB_Config(v_SCG_DRB_Id,
                                                       cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsDRB, cs_RlcBearerRouting_NR(p_MCG_PCellId)),
                                                       cs_NR_RlcBearerConfig_AM(v_IsFR1_SCG, -, f_NR_LogicalChannelId_DRB(v_SCG_DRB_Id)),
                                                       v_SDAP_ConfigurationSCG); //WA#WI=970303
    
    v_SS_TargetSCG_RadioBearerList := {v_SS_RadioBearerDRB_SCG};
    v_CellGroupConfigSCG := f_NR_GetSCGConfigNRDC_SCG(p_Target_PSCellId, v_SCG_DRB_Id, p_RNTI_Value);
    v_CellGroupConfigSCG.rlc_BearerToAddModList[0].reestablishRLC := true_;
    //@sic R5s221057 sic@
    v_RadioBearerConfig2 := cs_NRDRB_SplitBearer(omit, v_SCG_DRB_Id, cs_NR_PDCP_Config_Split(cs_38508_PDCP_Config_DRB, omit, tsc_NR_CellGroupId_SCG), secondary, f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get(), true_);
    v_V1530Ext := f_NRDC_BuildRRCReconfiguration_v1530( v_CellGroupConfigSCG, v_CellGroupConfigMCG, v_RadioBearerConfig2, -, p_SK_Counter );
    
    // Add SCG DRB configured in proxy (split SCG DRB)
    v_SS_MCG_RadioBearerList[lengthof(v_SS_MCG_RadioBearerList)] := cs_NR_SS_DRB_Config(v_SCG_DRB_Id, cs_NR_PDCP_Configuration_Proxy(cs_RlcBearerRouting_NR(p_Target_PSCellId)), cs_NR_RlcBearerConfig_AM(v_IsFR1_SCG, -, f_NR_LogicalChannelId_DRB(v_SCG_DRB_Id)),cs_SDAP_Configuration_None); //WA#WI=970303


    //1. NR Target PSCell: Configuration of SRB3 (if necessary) and DRBs
    //f_NR_SS_CommonRadioBearerConfig(p_Target_PSCellId, v_SS_TargetSCG_RadioBearerList); //WA#WI=970303
    
    // 2. Transfer of the PDCP Count for DRBs and SRB3 (if necessary) from NR source to NR target PSCell:
    //a) NR Source PSCell: Get PDCP COUNT.
    //b) NR Target PSCell: Set PDCP COUNT.
    //NOTE 1: No further sending/receiving of DRB data before the PSCell change is done.
    //NOTE 2: For AM DRBs the PDCP count is maintained. For SRB3 (if applied) and UM DRBs, the PDCP count is maintained or reset depending on the RRCReconfiguration message content.
    //WA#WI=zzz moved to be configured after split bearer release from source cell / otherwise we will have 2 split proxy mapped to MCG
    //f_NR_SS_PdcpCount_Handover(p_Source_PSCellId, p_Target_PSCellId, v_SS_TargetSCG_RadioBearerList, p_RBConfig_KeyChange);
    v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_Source_PSCellId, v_SS_TargetSCG_RadioBearerList, p_RBConfig_KeyChange);    //WA#WI=970303
    //3. NR Target PSCell: Inform the SS about the PSCell change and about the source cell id.
    f_NR_SS_PdcpHandoverCtrl(p_Target_PSCellId, cas_NR_PdcpHandoverInit_REQ(p_Source_PSCellId, p_Target_PSCellId));
        
    f_NR_SS_CommonRadioBearerConfig(p_Source_PSCellId, v_SourceSCG_RadioBearerToReleaseList); //WA#WI=970303
    //4. NR Target PSCell: Configure RACH procedure either dedicated or C-RNTI based. 
    //NOTE 3: The FollowOnFlag is set to true in the ASP reconfiguring C-RNTI.
    f_NR_CellInfo_SetRNTI(p_Target_PSCellId, p_RNTI_Value);
    f_NR_SS_C_RNTI_Config(p_Target_PSCellId, p_RNTI_Value, -, tsc_NoCnfReq, tsc_FollowOnFlag);
    f_NR_SS_RachProcedureConfigHO(p_Target_PSCellId, v_RachProcedureConfig);
    
    
    //5. NR Target PSCell: Activate security.
    if (isvalue(p_SK_Counter)) {
        v_SecurityParams.SK_Counter := valueof(p_SK_Counter);
        v_SecurityParams := f_NR_Security_InitSCG(v_SecurityParams);
        f_NR_Security_Set(v_SecurityParams);
    }
    //v_SecurityParams.AS_Ciphering_S.ActTimeList := f_NR_RRC_CipherActTime_SCG(p_Target_PSCellId, v_SS_TargetSCG_RadioBearerList);
    /*f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_Target_PSCellId,
                                          v_SecurityParams.AS_Integrity_S,
                                          v_SecurityParams.AS_Ciphering_S);
                                          */ //WA#WI=970303
    //6. NR Target PSCell: Configure UL grant configuration ("OnSR", default grant).
    //NOTE 3: Unless explicitly specified UL grant configuration keeps configured as per default at the NR source cell.
    f_NR_ULGrantConfiguration_Start(p_Target_PSCellId, -, -, -, -, cs_UL_GrantConfig_OnSR);
 
    //7. NR PCell: Send RRCReconfiguration.
    //8. NR PCell: Receive RRCReconfigurationComplete.
    v_Timing := f_NR_GetNextSendOccasion_DL(p_MCG_PCellId);
    f_NR_SendRRCReconfiguration (p_MCG_PCellId, tsc_NR_RbId_SRB1, -, v_V1530Ext, -, -, cs_TimingInfo_NR(v_Timing), false);
    
    //Reset RLC in MCG
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);
    f_NR_SS_SRBs_DRBs_Release(p_MCG_PCellId, cs_TimingInfo_NR(v_Timing), v_SS_DrbToReleaseMCG,-,tsc_FollowOnFlag); //WA#WI=970303 followonFlag added
    f_NR_SS_C_RNTI_Config(p_MCG_PCellId, tsc_C_RNTI_Value3, cs_TimingInfo_NR(v_Timing), tsc_NoCnfReq, tsc_FollowOnFlag); //WA#WI=970303 c-rnti added
    f_NR_SS_RachProcedureConfigHO(p_MCG_PCellId, v_RachProcedureConfig_MCG, cs_TimingInfo_NR(v_Timing), omit); //WA#WI=970303
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms
    f_NR_SS_SRBs_DRBs_Config(p_MCG_PCellId, cs_TimingInfo_NR(v_Timing), v_SS_MCG_RadioBearerList);
    //f_NR_SS_PdcpCount_Set(p_MCG_PCellId, v_PdcpCountInfoListMCG, cs_TimingInfo_NR(v_Timing)); //WA#WI=970303
    

    SRB.receive(car_NR_SRB_RrcPdu_IND(p_MCG_PCellId, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete(cr_RRCReconfigurationComplete_v1530(-, cr_RRCReconfigurationComplete_v1560(?)))));
    f_NR_SS_CommonRadioBearerConfig(p_Target_PSCellId, v_SS_TargetSCG_RadioBearerList); //WA#WI=970303
    f_NR_SS_PdcpCount_Set(p_Target_PSCellId, v_NR_PdcpCountInfoList); //WA#WI=970303
    //9. NR Target PSCell: Inform the SS about completion of the PSCell change (e.g. to trigger PDCP STATUS REPORT PDU).
    f_NR_SS_PdcpHandoverCtrl(p_Target_PSCellId, cas_NR_PdcpHandoverComplete_REQ(p_Target_PSCellId));

    //10. NR Source PSCell: Release SRB3 (if necessary) and DRBs.
    //f_NR_SS_CommonRadioBearerConfig(p_Source_PSCellId, v_SourceSCG_RadioBearerToReleaseList); //WA#WI=970303
    v_SecurityParams.AS_Ciphering_S.ActTimeList := f_NR_RRC_CipherActTime_SCG(p_Target_PSCellId, v_SS_TargetSCG_RadioBearerList); //WA#WI=970303
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_Target_PSCellId, //WA#WI=970303
                                          v_SecurityParams.AS_Integrity_S,
                                          v_SecurityParams.AS_Ciphering_S);
   } 


1.2 [bookmark: _Toc153775865]Change to the Function f_NR_GetRACHProcedureConfigPSCell
	Function name
	f_NR_GetRACHProcedureConfigPSCell

	Reason for change
	To do the reconfigurationWithSync rach configuration on the target PScell with newUE_Identity = (16712) , tsc_C_RNTI_Value3 is to be configured . so that it matches with peer reconfiguration configuration.
This is done by adding an extra parameter to the function f_NR_GetRACHProcedureConfigPSCell

	Summary of change
	p_RNTI_Value is added as a parameter to the function f_NR_GetRACHProcedureConfigPSCell

	TTCN module
	Common\NR\NR_HandoverCommonFunctions.ttcn

	MCC160 Comment
	Accepted



Before Change:
	      function f_NR_GetRACHProcedureConfigPSCell(NR_CellId_Type                       p_NR_CellId,
                                             template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit) runs on NR_BASE_PTC return template (value) NR_RachProcedureConfig_Type
  {
    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_NR_CellId);
        
    if (isvalue(p_RACH_ConfigDedicated.uplink)){
       v_RachProcedureConfig := cs_NR_RachProcedureConfig_CFRA(v_NR_PhysicalParameters);
    } else { //Default CBRA procedure
       // As per 38.523-3 Table 7.1.2.3.3-1: Default grants @sic R5-198994 sic@
       v_RachProcedureConfig := cs_NR_RachProcedureConfig_CBRA_CrntiBased(v_NR_PhysicalParameters, -, -, tsc_NR_ENDC_Msg3Grant_Def);
    }
    return v_RachProcedureConfig;
  }



After Change:
	  function f_NR_GetRACHProcedureConfigPSCell(NR_CellId_Type                       p_NR_CellId,
                                             template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,
                                             RNTI_Value_Type p_TempC_RNTI_Value := tsc_C_RNTI_Value2) runs on NR_BASE_PTC return template (value) NR_RachProcedureConfig_Type //WA#WI=970303
  {
    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_NR_CellId);
        
    if (isvalue(p_RACH_ConfigDedicated.uplink)){
       v_RachProcedureConfig := cs_NR_RachProcedureConfig_CFRA(v_NR_PhysicalParameters);
    } else { //Default CBRA procedure
       // As per 38.523-3 Table 7.1.2.3.3-1: Default grants @sic R5-198994 sic@
       v_RachProcedureConfig := cs_NR_RachProcedureConfig_CBRA_CrntiBased(v_NR_PhysicalParameters, -, p_TempC_RNTI_Value, tsc_NR_ENDC_Msg3Grant_Def); //WA#WI=970303
    }
    return v_RachProcedureConfig;
  }



